

```

<EntryPoint> -> <MainBlock>+
<MainBlock> -> <ApplicationBlock> | <ViewBlock>

<ApplicationBlock> ->
    application id indent <SubBlock>* detent

<SubBlock> -> <PropertiesBlock> | <VariablesBlock>
    | <MenuBlock> | <InitBlock> | <DrawBlock>
    | <OnKeyDownBlock> | <OnKeyUpBlock>
    | <OnKeyPressBlock> | <MethodBlock>

<ViewBlock> -> view id indent <SubBlock>* detent

<PropertiesBlock> ->
    properties indent <PropertyItem>* detent
<PropertyItem> -> defaultview = id EOL
    | uid = intnbr EOL
    | refreshrate = intnbr EOL

<VariablesBlock> ->
    variables indent <Declarations>+ detent
<Declarations> -> public indent <DeclarationLine>+ detent
    | private indent <DeclarationLine>+ detent
    | <Visibility> <DeclarationLine>

<DeclarationLine> -> <Declaration> EOL
<Declaration> -> <TypeDef> <IdList>
<TypeDef> -> type <ArrayDef>+

<IdList> -> id <IdListMore>
<IdListMore> -> , id <IdListMore> |  $\epsilon$ 

<ArrayDef> -> [ <Size> ]
<Size> -> intnbr |  $\epsilon$ 

<MenuBlock> -> menu indent <MainMenuItem>* detent
<MainMenuItem> -> <MenuItem> <SubMenuBlock>
<MenuItem> -> id : stringliteral EOL
<SubMenuBlock> -> indent <MenuItem>* detent |  $\epsilon$ 

<InitBlock> -> init <StatementsBlock>

```

```

<DrawBlock> -> draw <StatementsBlock>
<OnKeyDownBlock> -> onkeydown <StatementsBlock>
<OnKeyUpBlock> -> onkeyup <StatementsBlock>
<OnKeyPressBlock> -> onkeypress <StatementsBlock>

<MethodBlock> -> <MethodDef> <StatementsBlock>
<MethodDef> -> <Visibility> <Type> id ( <ParameterDefs> )
<Visibility> -> private | public |  $\epsilon$ 
<Type> -> type |  $\epsilon$ 

<ParameterDefs> -> <TypeDef> id <ParameterDefsMore> |  $\epsilon$ 
<ParameterDefsMore> ->
    , <TypeDef> id <ParameterDefsMore>
    |  $\epsilon$ 

<StatementsBlock> -> Indent <Statements> detent | EOL
<Statements> -> <Statement> <Statements> |  $\epsilon$ 
<Statement> -> <SimpleStatement> EOL | <Declaration> EOL
    | <IfStatement> | <WhileStatement>

<SimpleStatement> ->
    <ImportedStatement> | <LoadBitmapStatement>
    | <Allocation> | <MethodCall> | <ReturnStatement>

<ImportedStatement> ->
    importedstatement ( <Parameters> )
<LoadBitmapStatement> -> loadbitmap ( stringliteral )

<Parameters> -> <Expression> <ParametersMore> |  $\epsilon$ 
<ParametersMore> -> , <Expression> <ParametersMore> |  $\epsilon$ 

<Expression> -> <LogicTerm> <RestOfExpression>

<RestOfExpression> ->
    or <LogicTerm> <RestOfExpression>
    |  $\epsilon$ 
<LogicTerm> -> <RelExpression> <RestOfLogicTerm>
<RestOfLogicTerm> -> and <RelExpression> <RestOfLogicTerm>
    |  $\epsilon$ 

<RelExpression> ->
    <SimpleExpression> <RestOfRelExpression>
<RestOfRelExpression> ->

```

```

    <RelationOperator> <SimpleExpression> | ε
<RelationOperator> -> = | <= | >= | < | > | <>

<SimpleExpression> -> <Sign> <Term> <RestOfSimpleExpression>
<Sign> -> + | - | ε

<RestOfSimpleExpression> ->
    <AddOperator> <Term> <RestOfSimpleExpression>
    | ε

<Term> -> <FactorNeg> <RestOfTerm>
<RestOfTerm> ->
    <MultiplicationOperator> <Factor> <RestOfTerm>
    | ε

<FactorNeg> -> not <Factor> | <Factor>

<Factor> -> ( <Expression> )
    | <Operand>
<AddOperator> -> + | -
<MultiplicationOperator> -> * | / | %

<Operand> -> <Variable> | <MethodCall>
    | <ImportedStatement> | < LoadBitmapStatement >
    | enum | intnbr | realnbr | stringliteral
    | true | false

<Variable> -> id <Array>*
<Array> -> [ <Expression> ]

<Allocation> -> <Variable> <Assignment> <Expression>
<Assignment> -> = | += | -= | *= | /= | %=

<MethodCall> -> id ( <Parameters> )

<IfStatement> -> if ( <Expression> ) <Then>

<Then> -> <SimpleStatement> EOL <Else>
    | <StatementsBlock> <Else>

<Else> -> else <ElseThen> | ε
<ElseThen> -> <SimpleStatement> EOL
    | <StatementsBlock>
    | <IfStatement>

```

<WhileStatement> -> **while** (<Expression>) <WhileDo>

<WhileDo> -> <SimpleStatement> **EOL**

| <StatementsBlock>

<ReturnStatement> -> **return** <Expression>